
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of EMM test case 9.2.1.2.13 which are part of the LTE test suite.

The test case can be demonstrated to run with two LTE UE (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Cases:
tc_9_2_1_2_13
Test Group:
EMM

ATS Version:
iwd-EUTRA-TVB2009-12_D11wk04
System Simulator used:
Anite Conformance Toolset
UE(s) used:                       Qualcomm 9600 UE

Verification Status:
PASS
4 Corrections required for test case 9.2.1.2.13

4.1 Introduction

This section describes the changes required to make test case 9.2.1.2.13 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-03_D11wk04 release.

4.2 Change 1

Testcase name
9.2.1.2.13

Protocol
EMM

Function Name
f_TC_9_2_1_2_13_EUTRA()

Reason for change
1. UTRA/GERAN PTC should wait for EUTRA PTC to configure UE in combined attach mode before performing pre-preamble. 

2. Preamble should be performed on Cell I, as Cell A is not configured in this test case.

Note: Prose CR will be raised for this change at the next RAN5 meeting 

Summary of change
1. Sent the IRAT co-ordination trigger to Other PTC after configuring UE in combined attach mode

2. Replaced Cell A with Cell I in preamble.

Source of change
File : NAS_Attach_UG.ttcn

Before:

Source: NAS_Attach_UG.ttcn

function f_TC_9_2_1_2_13_EUTRA() runs on EUTRA_PTC

{

    var boolean v_TriggerRxd := false;

    if (pc_UTRA) {

      f_EUTRA_NAS_Init(LTE_CellsOnDifferentPLMN, c4, LTE_NAS_Usef1_f2);

    } else if (pc_GERAN) {

      f_EUTRA_NAS_Init(LTE_CellsOnDifferentPLMN, c5, LTE_NAS_Usef1_f2);

    } else { // Neither UMTS, nor GSM, Test purpose can't be checked

       f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UTRAN or GERAN are mandatory for this test case");

    }

    // create cell and send out system information

    f_EUTRA_CellConfig_Def(eutra_CellI);

    f_EUTRA_CellConfig_SRBs_Only(eutra_CellJ);

    f_EUTRA_CellConfig_SRBs_Only(eutra_CellK);

    f_EUTRA_CellConfig_Def(eutra_CellL);

    f_UT_ConfigureCombinedAttach (UT); // R5-103815

    // Preamble: UE is in state Switched OFF (state 1)

    // ==========================================================================

    // If pc_GERAN or pc_UTRAN supported by UE, wait for other RAT to update TMSI-1, P-TMSI-1and RAI-1, and GPRS update status is "GU1:UPDATED"

    if (pc_UTRA) {

        v_TriggerRxd := f_WaitForIRATCoOrd_Trigger(UTRAN); // wait for trigger from UTRAN side

        f_EUTRA_Preamble_InterRAT(eutra_CellA, STATE2_IDLEUPDATE, UTRAN);

    } else if (pc_GERAN) {

        v_TriggerRxd := f_WaitForIRATCoOrd_Trigger(GERAN); // wait for trigger from GERAN side

        f_EUTRA_Preamble_InterRAT(eutra_CellA, STATE2_IDLEUPDATE, GERAN);

    } else { // supports neither UTRAN or GERAN

        f_EUTRA_Preamble(eutra_CellA, STATE2_IDLEUPDATE);

    }

    f_EUTRA_UE_Detach_SwitchOffUe ( eutra_CellI, RRC_IDLE );

    f_EUTRA_TestBody_Set(true);

    fl_TC_9_2_1_2_13_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellI, E2_CONNECTED);

} // end f_TC_9_2_1_2_13_EUTRA

After: NAS_Attach_UG.ttcn
function f_TC_9_2_1_2_13_EUTRA() runs on EUTRA_PTC

{

    ..

    var boolean v_TriggerRxd := false;

    if (pc_UTRA) {

      f_EUTRA_NAS_Init(LTE_CellsOnDifferentPLMN, c4, LTE_NAS_Usef1_f2);

    } else if (pc_GERAN) {

      f_EUTRA_NAS_Init(LTE_CellsOnDifferentPLMN, c5, LTE_NAS_Usef1_f2);

    } else { // Neither UMTS, nor GSM, Test purpose can't be checked

       f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UTRAN or GERAN are mandatory for this test case");

    }

    // create cell and send out system information

    f_EUTRA_CellConfig_Def(eutra_CellI);

    f_EUTRA_CellConfig_SRBs_Only(eutra_CellJ);

    f_EUTRA_CellConfig_SRBs_Only(eutra_CellK);

    f_EUTRA_CellConfig_Def(eutra_CellL);

    f_UT_ConfigureCombinedAttach (UT); // R5-103815

    f_SendIRATCoOrd (UTRAN, cms_IRAT_Trigger); //Anite change 1.1

    // Preamble: UE is in state Switched OFF (state 1)

    // ==========================================================================

    // If pc_GERAN or pc_UTRAN supported by UE, wait for other RAT to update TMSI-1, P-TMSI-1and RAI-1, and GPRS update status is "GU1:UPDATED"

    if (pc_UTRA) {

        v_TriggerRxd := f_WaitForIRATCoOrd_Trigger(UTRAN); // wait for trigger from UTRAN side

        f_EUTRA_Preamble_InterRAT(eutra_CellI, STATE2_IDLEUPDATE, UTRAN); //Anite change 1.2

    } else if (pc_GERAN) {

        v_TriggerRxd := f_WaitForIRATCoOrd_Trigger(GERAN); // wait for trigger from GERAN side

        f_EUTRA_Preamble_InterRAT(eutra_CellI, STATE2_IDLEUPDATE, GERAN); //Anite change 1.2

    } else { // supports neither UTRAN or GERAN

        f_EUTRA_Preamble(eutra_CellI, STATE2_IDLEUPDATE); //Anite change 1.2

    }

    f_EUTRA_UE_Detach_SwitchOffUe ( eutra_CellI, RRC_IDLE );

    f_EUTRA_TestBody_Set(true);

    fl_TC_9_2_1_2_13_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellI, E2_CONNECTED);

} // end f_TC_9_2_1_2_13_EUTRA

4.3 Change 2

Testcase name
9.2.1.2.13

Protocol
EMM

Function Name
fl_TC_9_2_1_2_13_Body()

Reason for change
1. Cell power states of all the cells should as per the Prose CR R5-110877.

2. At step 8,12 and 15, cell L power level should be configured as that of a  Suitable Neighbour intra frequency cell. 

3. At step 9, Attach Request message should be accepted without any integrity protection as security context is deleted at step 5.

4. At step 14, cellC is wrongly used. It should be cellJ

5. At step 18, Attach Request message should be accepted without any integrity protection as security context is not present in UE at this step. 

6. At the end of test case body, EUTRA PTC should send a trigger to other PTC to indicate end of TC

Summary of change
1. Changed the cell power states of the cells as per the Prose CR R5-110877.
2. At step 8,12 and 15, cell L power level configured as that of a  Suitable Neighbour intra frequency cell. 
3. At step 9, accepted attach request without any integrity protection.

4. At step 14, Replaced CellC with Cell J.

5. At step 18, accepted attach request without any integrity protection.

6. At the end of test case body, sent an IRAT coordination trigger to other PTC.



Source of change
File : NAS_Attach_UG.ttcn

Before:

Source: NAS_Attach_UG.ttcn

    function fl_TC_9_2_1_2_13_Body() runs on EUTRA_PTC

    {

        var boolean v_TriggerRxd := false;

        var template (value) CellPowerList_Type v_CellPowerList;

        var NAS_MSG_Indication_Type v_NAS_Ind;

        var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType();

        var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

        var charstring v_OtherPTC := "";

        if (pc_UTRA) {

           v_OtherPTC := "UTRAN";

         } else if (pc_GERAN) {

           v_OtherPTC := "GERAN";

         } else { // Neither UMTS, nor GSM, so just do LTE

           v_OtherPTC := "NONE";

        }

        // Set Power levels

        //@siclog "Step 1" siclog@

        v_CellPowerList := {

          cs_CellPower ( eutra_CellI, tsc_ServingCellRS_EPRE ),

          cs_CellPower ( eutra_CellK, tsc_SuitableCellRS_EPRE ),

          cs_CellPower ( eutra_CellL, tsc_NonSuitableCellRS_EPRE ),

          cs_CellPower ( eutra_CellJ, tsc_NonSuitableCellRS_EPRE )

        };

        f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

        //@siclog "Step 2 - 5" siclog@

        // UE is switched on, sends and ATTACH REQUEST and SS transmits an ATTACH REJECT with cause set to

        // No suitable cells in this tracking area

        f_EUTRA_AttachReject (eutra_CellI, tsc_Cause_NoSuitableCells);

        //  Set Power levels

        //@siclog "Step 6" siclog@

        v_CellPowerList := {

          cs_CellPower ( eutra_CellI, tsc_SuitableCellRS_EPRE ),

          cs_CellPower ( eutra_CellK, tsc_ServingCellRS_EPRE ),

          cs_CellPower ( eutra_CellL, tsc_NonSuitableCellRS_EPRE ),

          cs_CellPower ( eutra_CellJ, tsc_NonSuitableCellRS_EPRE )  

        };

        f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

        //@siclog "Step 7" siclog@

        //   Check: Does the UE transmit an ATTACH REQUEST message in the next 30 seconds on Cell K or Cell I?

        f_EUTRA_CheckNoAttachOn2Cells(eutra_CellK, eutra_CellI, 30.0, "step 7, ATTACH REQUEST message on Cell K or Cell I");

        //@siclog "Step 8" siclog@

        //  Set Power levels

        v_CellPowerList := {

          cs_CellPower ( eutra_CellI, tsc_NonSuitableCellRS_EPRE ),

          cs_CellPower ( eutra_CellK, tsc_ServingCellRS_EPRE ),

          cs_CellPower ( eutra_CellL, tsc_SuitableCellRS_EPRE ), 

          cs_CellPower ( eutra_CellJ, tsc_SuitableCellRS_EPRE )

        };

        f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

        //@siclog "Step 9" siclog@

        // Check: Does the UE transmit an ATTACH REQUEST message including a PDN CONNECTIVITY REQUEST message as specified on Cell L?

        v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellL,

                                                    tsc_RRC_TI_Def,

                                                    cr_EstablishmentCause_Any,

                                                    cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                                                  cdr_ATTACH_REQUEST_WithIMSIAndnoTAI(

                                                                                       v_AttachType,

                                                                                       v_AdditionalUpdateType,

                                                                                       f_Imsi2MobileIdentity( px_IMSI_Def ),

                                                                                       cr_TMSI_Status('0'B)),

                                                                                  cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                       f_GetPdnType())));

        f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 9");

        //@siclog "Step 10" siclog@

        // The SS transmits an ATTACH REJECT message with the EMM cause set to

        // No suitable cells in this tracking area on cell L

        //(The list of "forbidden tracking areas for roaming" in the UE should now contain TAI-9 and TAI-11)

        SRB.send  ( cas_SRB1_NasPdu_REQ ( eutra_CellL,

                                          cs_TimingInfo_Now,

                                          cs_NAS_Request ( tsc_SHT_NoSecurityProtection,

                                                         cs_508_AttachReject ( cs_EMM_Cause_v(tsc_Cause_NoSuitableCells),

                                                                               omit))));

        //  The SS releases the RRC connection.

        //@siclog "Step 11" siclog@

        f_EUTRA_RRC_ConnectionRelease(eutra_CellL, tsc_RRC_TI_Def, other);

        // Set Power levels

        //@siclog "Step 12" siclog@

        v_CellPowerList := {

          cs_CellPower ( eutra_CellI, tsc_ServingCellRS_EPRE ),

          cs_CellPower ( eutra_CellK, tsc_NonSuitableCellRS_EPRE ),

          cs_CellPower ( eutra_CellL, tsc_SuitableCellRS_EPRE ), 

          cs_CellPower ( eutra_CellJ, tsc_NonSuitableCellRS_EPRE )

        };

        f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

        //@siclog "Step 13" siclog@

        // Check: Does the UE transmit the ATTACH REQUEST message in the next 30 seconds on CellI?

        f_EUTRA_CheckNoAttachOn1Cell(eutra_CellI, 30.0, "step 13");

        //  Configure Cells.

        //@siclog "Step 14" siclog@

        v_CellPowerList := {

          cs_CellPower ( eutra_CellI, tsc_NonSuitableCellRS_EPRE ),

          cs_CellPower ( eutra_CellK, tsc_NonSuitableCellRS_EPRE ),

          cs_CellPower ( eutra_CellL, tsc_NonSuitableCellRS_EPRE ),

          cs_CellPower ( eutra_CellC, tsc_NonSuitableCellRS_EPRE )

          }

        select (v_OtherPTC) { // change power levels, send trigger to other RAT, then wait for trigger to be returned

          case ("UTRAN") {

            f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

            f_SendIRATCoOrd (UTRAN, cms_IRAT_Trigger);

            v_TriggerRxd := f_WaitForIRATCoOrd_Trigger(UTRAN); // wait for trigger from UTRAN side

          } case ("GERAN") {

            f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

            f_SendIRATCoOrd (GERAN, cms_IRAT_Trigger);

            v_TriggerRxd := f_WaitForIRATCoOrd_Trigger(GERAN); // wait for trigger from GERAN side

          } case else {} // Do nothing

        }

        //  Set Power levels. Cell5/24 comes from Non-Suitable Cell level

        //@siclog "Step 15" siclog@

        v_CellPowerList := {

          cs_CellPower ( eutra_CellI, tsc_ServingCellRS_EPRE ),

          cs_CellPower ( eutra_CellK, tsc_NonSuitableCellRS_EPRE ),

          cs_CellPower ( eutra_CellL, tsc_SuitableCellRS_EPRE ),

          cs_CellPower ( eutra_CellJ, tsc_NonSuitableCellRS_EPRE )

        };

        f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

        //@siclog "Step 16" siclog@

        // If possible (see ICS) switch off is performed. Otherwise the power is removed.

        f_UT_SwitchOffUE (UT, false);

        //@siclog "Step 17" siclog@

        // The UE is brought back to operation.

        f_UT_SwitchOnUE (UT, false );

        //@siclog "Step 18" siclog@

        // Check: Does the UE transmit an ATTACH REQUEST message including a PDN CONNECTIVITY REQUEST message?

        v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellI,

                                                    tsc_RRC_TI_Def,

                                                    cr_EstablishmentCause_Any,

                                                    cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                                                  cdr_ATTACH_REQUEST_WithIMSIAndnoTAI(

                                                                                       v_AttachType,

                                                                                       v_AdditionalUpdateType,

                                                                                       f_Imsi2MobileIdentity( px_IMSI_Def ),

                                                                                       cr_TMSI_Status('0'B)),

                                                                                  cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                       f_GetPdnType())));

        f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 18");

        //@siclog "Step 19 - 30" siclog@

        f_EUTRA_IdleUpdated_Step5_17 (eutra_CellI, STATE2_IDLEUPDATE, NORMAL, v_NAS_Ind, false);

    }// end function fl_TC_9_2_1_2_13_Body

After: NAS_Attach_UG.ttcn
    function fl_TC_9_2_1_2_13_Body() runs on EUTRA_PTC

    {

        var boolean v_TriggerRxd := false;

        var template (value) CellPowerList_Type v_CellPowerList;

        var NAS_MSG_Indication_Type v_NAS_Ind;

        var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType();

        var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

        var charstring v_OtherPTC := "";

        if (pc_UTRA) {

           v_OtherPTC := "UTRAN";

         } else if (pc_GERAN) {

           v_OtherPTC := "GERAN";

         } else { // Neither UMTS, nor GSM, so just do LTE

           v_OtherPTC := "NONE";

        }

        // Set Power levels

        //@siclog "Step 1" siclog@

        v_CellPowerList := {

          cs_CellPower ( eutra_CellI, tsc_ServingCellRS_EPRE ),

          cs_CellPower ( eutra_CellK, tsc_SuitableCellRS_EPRE ),

          cs_CellPower ( eutra_CellL, tsc_NonSuitableCellRS_EPRE ), 

          cs_CellPower ( eutra_CellJ, tsc_NonSuitableOffCellRS_EPRE )  //Anite change 2.1

        };

        f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

        //@siclog "Step 2 - 5" siclog@

        // UE is switched on, sends and ATTACH REQUEST and SS transmits an ATTACH REJECT with cause set to

        // No suitable cells in this tracking area

        f_EUTRA_AttachReject (eutra_CellI, tsc_Cause_NoSuitableCells);

        //  Set Power levels

        //@siclog "Step 6" siclog@

        v_CellPowerList := {

          cs_CellPower ( eutra_CellI, tsc_SuitableCellRS_EPRE ),

          cs_CellPower ( eutra_CellK, tsc_ServingCellRS_EPRE ),

          cs_CellPower ( eutra_CellL, tsc_NonSuitableCellRS_EPRE ), 

          cs_CellPower ( eutra_CellJ, tsc_NonSuitableOffCellRS_EPRE ) //Anite change 2.1

        };

        f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

        //@siclog "Step 7" siclog@

        //   Check: Does the UE transmit an ATTACH REQUEST message in the next 30 seconds on Cell K or Cell I?

        f_EUTRA_CheckNoAttachOn2Cells(eutra_CellK, eutra_CellI, 30.0, "step 7, ATTACH REQUEST message on Cell K or Cell I");

        //@siclog "Step 8" siclog@

        //  Set Power levels

        v_CellPowerList := {

          cs_CellPower ( eutra_CellI, tsc_NonSuitableOffCellRS_EPRE ), //Anite change 2.1

          cs_CellPower ( eutra_CellK, tsc_ServingCellRS_EPRE ), 

          cs_CellPower ( eutra_CellL, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE ), //Anite change 2.2

          cs_CellPower ( eutra_CellJ, tsc_SuitableCellRS_EPRE )

        };

        f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

        //@siclog "Step 9" siclog@

        // Check: Does the UE transmit an ATTACH REQUEST message including a PDN CONNECTIVITY REQUEST message as specified on Cell L?

        v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellL,

                                                    tsc_RRC_TI_Def,

                                                    cr_EstablishmentCause_Any,

                                                    cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection, //anite change 2.3

                                                                                  cdr_ATTACH_REQUEST_WithIMSIAndnoTAI(

                                                                                       v_AttachType,

                                                                                       v_AdditionalUpdateType,

                                                                                       f_Imsi2MobileIdentity( px_IMSI_Def ),

                                                                                       cr_TMSI_Status('0'B)),

                                                                                  cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                       f_GetPdnType())));

        f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 9");

        //@siclog "Step 10" siclog@

        // The SS transmits an ATTACH REJECT message with the EMM cause set to

        // No suitable cells in this tracking area on cell L

        //(The list of "forbidden tracking areas for roaming" in the UE should now contain TAI-9 and TAI-11)

        SRB.send  ( cas_SRB1_NasPdu_REQ ( eutra_CellL,

                                          cs_TimingInfo_Now,

                                          cs_NAS_Request ( tsc_SHT_NoSecurityProtection,

                                                         cs_508_AttachReject ( cs_EMM_Cause_v(tsc_Cause_NoSuitableCells),

                                                                               omit))));

        //  The SS releases the RRC connection.

        //@siclog "Step 11" siclog@

        f_EUTRA_RRC_ConnectionRelease(eutra_CellL, tsc_RRC_TI_Def, other);

        // Set Power levels

        //@siclog "Step 12" siclog@

        v_CellPowerList := {

          cs_CellPower ( eutra_CellI, tsc_ServingCellRS_EPRE ),

          cs_CellPower ( eutra_CellK, tsc_NonSuitableCellRS_EPRE ), 

          cs_CellPower ( eutra_CellL, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE ), //Anite change 2.2

          cs_CellPower ( eutra_CellJ, tsc_NonSuitableOffCellRS_EPRE )

        };

        f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

        //@siclog "Step 13" siclog@

        // Check: Does the UE transmit the ATTACH REQUEST message in the next 30 seconds on CellI?

        f_EUTRA_CheckNoAttachOn1Cell(eutra_CellI, 30.0, "step 13");

        //  Configure Cells.

        //@siclog "Step 14" siclog@

        v_CellPowerList := {

          cs_CellPower ( eutra_CellI, tsc_NonSuitableCellRS_EPRE ),

          cs_CellPower ( eutra_CellK, tsc_NonSuitableOffCellRS_EPRE ), //Anite change 2.1

          cs_CellPower ( eutra_CellL, tsc_NonSuitableCellRS_EPRE ),

          cs_CellPower ( eutra_CellJ, tsc_NonSuitableOffCellRS_EPRE ) //Anite change 2.4

          }

        select (v_OtherPTC) { // change power levels, send trigger to other RAT, then wait for trigger to be returned

          case ("UTRAN") {

            f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

            f_SendIRATCoOrd (UTRAN, cms_IRAT_Trigger);

            v_TriggerRxd := f_WaitForIRATCoOrd_Trigger(UTRAN); // wait for trigger from UTRAN side

          } case ("GERAN") {

            f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

            f_SendIRATCoOrd (GERAN, cms_IRAT_Trigger);

            v_TriggerRxd := f_WaitForIRATCoOrd_Trigger(GERAN); // wait for trigger from GERAN side

          } case else {} // Do nothing

        }

        //  Set Power levels. Cell5/24 comes from Non-Suitable Cell level

        //@siclog "Step 15" siclog@

        v_CellPowerList := {

          cs_CellPower ( eutra_CellI, tsc_ServingCellRS_EPRE ),

          cs_CellPower ( eutra_CellK, tsc_NonSuitableOffCellRS_EPRE ), //Anite change 2.1

          cs_CellPower ( eutra_CellL, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE), //anite change 2.2

          cs_CellPower ( eutra_CellJ, tsc_NonSuitableOffCellRS_EPRE ) //Anite change 2.1

        };

        f_EUTRA_SetCellPowerList ( v_CellPowerList, cs_TimingInfo_Now );

        //@siclog "Step 16" siclog@

        // If possible (see ICS) switch off is performed. Otherwise the power is removed.

        f_UT_SwitchOffUE (UT, false);

        //@siclog "Step 17" siclog@

        // The UE is brought back to operation.

        f_UT_SwitchOnUE (UT, false );

        //@siclog "Step 18" siclog@

        // Check: Does the UE transmit an ATTACH REQUEST message including a PDN CONNECTIVITY REQUEST message?

        v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellI,

                                                    tsc_RRC_TI_Def,

                                                    cr_EstablishmentCause_Any,

                                                    cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection, //anite change 2.5

                                                                                  cdr_ATTACH_REQUEST_WithIMSIAndnoTAI(

                                                                                       v_AttachType,

                                                                                       v_AdditionalUpdateType,

                                                                                       f_Imsi2MobileIdentity( px_IMSI_Def ),

                                                                                       cr_TMSI_Status('0'B)),

                                                                                  cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                       f_GetPdnType())));

        f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 18");

        //@siclog "Step 19 - 30" siclog@

        f_EUTRA_IdleUpdated_Step5_17 (eutra_CellI, STATE2_IDLEUPDATE, NORMAL, v_NAS_Ind, false);

        select (v_OtherPTC) { //Anite change 2.6

          case ("UTRAN") {

            f_SendIRATCoOrd (UTRAN, cms_IRAT_Trigger);

          } case ("GERAN") {

            f_SendIRATCoOrd (GERAN, cms_IRAT_Trigger);

          } case else {} // Do nothing

        }
    }// end function fl_TC_9_2_1_2_13_Body

4.4 Change 3

Testcase name
9.2.1.2.13

Protocol
EMM

Function Name
f_TC_9_2_1_2_13_UTRAN ()

Reason for change
1. Variable v_plmn should be initialised after UTRA INIT.

2. Before preamble, UTRAN PTC should wait for a trigger from EUTRA PTC after the UE is configured in combined attach mode. 

3. Cell5 should be set to switch off immediately after preamble and there is no change in cell power at step 1 as per test prose.

4. At step14a3 RRC Conn Est should be as performed on Cell5.

5.  RRC Connection Release should be performed on Cell5 and should be done immediately after step14a4.

6. After step15 , UTRAN PTC should wait for trigger from EUTRA PTC to indicate the end of test case.

7. Post amble is performed in EUTRA component.

Summary of change
1. Initialised v_plmn after preamble.
2. Introduced a wait for trigger from EUTRA PTC before preamble in UTRAN.
3. Cell power of cell5 is set to tsc_AttenuationCellOFF after preamble.
4. At step14a3 replaced utran_CellDedicated with utran_Cell5.
5. Moved RRC Connection Release after step 14a4.
6. After step 14, introduced a wait for trigger from EUTRA PTC to indicate end of test case. 
7. Commented post amble and released cell5 after test body.

Source of change
File : NAS_Attach_UG_UTRAN.ttcn

Before:

Source: NAS_Attach_UG_UTRAN.ttcn

function f_TC_9_2_1_2_13_UTRAN () runs on UTRAN_PTC

{

  var boolean v_TriggerRxd := false;

  var RRC_DATA_IND v_RRC_DataInd;

  var octetstring v_PLMN := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran_Cell5));
  //var O1_Type v_RAC := f_UTRAN_CellInfo_GetRAC(utran_Cell5);

  //var O2_Type v_LAC := f_UTRAN_CellInfo_GetLAC(utran_Cell5);

  if (pc_UTRA) { // UE supports UTRAN

    f_UTRAN_Init(EUTRA_UTRAN); // This is still only a 2 PTC test

    f_UTRAN_SS_CreateCellDCH (utran_Cell5);

    f_UTRAN_SendDefSysInfo (utran_Cell5);

    // Pre-Preamble on UTRAN cell first to set PTSMI, PTSMI Sig and TMSI in USIM

    f_UTRAN_Pre_Preamble (utran_Cell5);

    f_UTRAN_InterRAT_InitialiseAuthParams();  // The UE starts in EUTRA, so wait for it to send the Auth Params

    f_UTRAN_TestBody_Set( true );

    // First Trigger

    //@siclog "Step 1" siclog@

    // Set Cell 5 to "Non Suitable Cell"

    f_UTRAN_SetCellPower (utran_Cell5, tsc_AttenuationNonSuitableCell);

    // Second Trigger

    //@siclog "Step 14a1" siclog@

    // Set Cell 5 to "Serving Cell"

    v_TriggerRxd := f_WaitForIRATCoOrd_Trigger (EUTRA);

    f_UTRAN_SetCellPower (utran_Cell5, tsc_AttenuationServingCell);

    //@siclog "Step 14a3" siclog@

    // Check: Does the UE transmit an ATTACH REQUEST message without P-TMSI, P-TMSI signature, RAI, TMSI

    f_UTRAN_RRC_ConnEst(utran_CellDedicated); // Establish RRC connection

    alt {

      [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,

                                                tsc_RB3,

                                                cr_U_AttachReqIMSInoTMSI (cr_AttachType(?, tsc_I_AttachTypeCombined),

                                                                          f_Imsi2MobileIdentity(px_IMSI_Def),

                                                                          cr_RAI (omit, cr_LAI(omit, v_PLMN, 'FFFE'O), ?))))

        -> value v_RRC_DataInd

        {

          f_UTRAN_PreliminaryPass (__FILE__, __LINE__, "Step 14a3");

        }

    }

    //@siclog "Step 14a4" siclog@

    // Send an ATTACH REJECT message with GMM Cause "No suitable cells in location area"

    U_Dc.send(cs_DataReq_PS(utran_CellDedicated,

                                tsc_RB3,

                                cs_U_AttachReject('0F'O)  // cause "No suitable cells in location area"

                                              ));

    //@siclog "Step 15" siclog@

    // Set Cell 5 to "Non Suitable Cell"

    f_UTRAN_SetCellPower (utran_Cell5, tsc_AttenuationNonSuitableCell);

    // Finished here for now

    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);

    f_UTRAN_RRC_ConnRel(utran_CellDedicated, cell_Dch);
    // Test case now finished on UTRAN side

    f_UTRAN_TestBody_Set( false );

    f_UTRAN_Postamble ( utran_Cell5, U1_IDLE );
    //  f_UTRAN_GMM_DetachOnSwitchOff (utran_Cell5);

    //  f_UTRAN_ReleaseCell ( utran_Cell5 );
  } // UE doesn't support UTRAN, so do nothing

} // f_TC_9_2_1_2_13_UTRAN

After: NAS_Attach_UG_UTRAN.ttcn
function f_TC_9_2_1_2_13_UTRAN () runs on UTRAN_PTC

{

  var boolean v_TriggerRxd := false;

  var RRC_DATA_IND v_RRC_DataInd;

  //var O1_Type v_RAC := f_UTRAN_CellInfo_GetRAC(utran_Cell5);

  //var O2_Type v_LAC := f_UTRAN_CellInfo_GetLAC(utran_Cell5);

  if (pc_UTRA) { // UE supports UTRAN

    f_UTRAN_Init(EUTRA_UTRAN); // This is still only a 2 PTC test

    f_UTRAN_SS_CreateCellDCH (utran_Cell5);

    f_UTRAN_SendDefSysInfo (utran_Cell5);

    var octetstring v_PLMN := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran_Cell5)); //Anite change
    // Pre-Preamble on UTRAN cell first to set PTSMI, PTSMI Sig and TMSI in USIM

    v_TriggerRxd := f_WaitForIRATCoOrd_Trigger (EUTRA); //Anite JD
    f_UTRAN_Pre_Preamble (utran_Cell5);

    f_UTRAN_InterRAT_InitialiseAuthParams();  // The UE starts in EUTRA, so wait for it to send the Auth Params

    f_UTRAN_TestBody_Set( true );

    // First Trigger

    //@siclog "Step 1" siclog@

    // Set Cell 5 to "Non Suitable Cell"

    f_UTRAN_SetCellPower (utran_Cell5, tsc_AttenuationCellOFF); //Anite JD

    // Second Trigger

    //@siclog "Step 14a1" siclog@

    // Set Cell 5 to "Serving Cell"

    v_TriggerRxd := f_WaitForIRATCoOrd_Trigger (EUTRA);

    f_UTRAN_SetCellPower (utran_Cell5, tsc_AttenuationServingCell);

    //@siclog "Step 14a3" siclog@

    // Check: Does the UE transmit an ATTACH REQUEST message without P-TMSI, P-TMSI signature, RAI, TMSI

    f_UTRAN_RRC_ConnEst(utran_Cell5); //Anite // Establish RRC connection

    alt {

      [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,

                                                tsc_RB3,

                                                cr_U_AttachReqIMSInoTMSI (cr_AttachType(?, tsc_I_AttachTypeCombined),

                                                                          f_Imsi2MobileIdentity(px_IMSI_Def),

                                                                          cr_RAI (omit, cr_LAI(omit, v_PLMN, 'FFFE'O), ?))))

        -> value v_RRC_DataInd

        {

          f_UTRAN_PreliminaryPass (__FILE__, __LINE__, "Step 14a3");

        }

    }

    //@siclog "Step 14a4" siclog@

    // Send an ATTACH REJECT message with GMM Cause "No suitable cells in location area"

    U_Dc.send(cs_DataReq_PS(utran_CellDedicated,

                                tsc_RB3,

                                cs_U_AttachReject('0F'O)  // cause "No suitable cells in location area"

                                              ));

    f_UTRAN_RRC_ConnRel(utran_Cell5, cell_Dch); //Anite change
    //@siclog "Step 15" siclog@

    // Set Cell 5 to "Non Suitable Cell"

    f_UTRAN_SetCellPower (utran_Cell5, tsc_AttenuationNonSuitableCell); 

    // Finished here for now

    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);

    v_TriggerRxd := f_WaitForIRATCoOrd_Trigger (EUTRA);//anite change
    // Test case now finished on UTRAN side

    f_UTRAN_TestBody_Set( false );

    //f_UTRAN_Postamble ( utran_Cell5, U1_IDLE );
    //  f_UTRAN_GMM_DetachOnSwitchOff (utran_Cell5);

    f_UTRAN_ReleaseCell ( utran_Cell5 ); //anite JD : was commented earlier.
  } // UE doesn't support UTRAN, so do nothing

} // f_TC_9_2_1_2_13_UTRAN



4.5 Change 4

Testcase name
9.2.1.2.13

Protocol
EMM

Function Name
f_TC_9_2_1_2_13_GERAN ()

Reason for change
1. Variable v_plmn should be initialised after UTRA INIT.

2. Before preamble, GERAN PTC should wait for a trigger from EUTRA PTC after the UE is configured in combined attach mode.

3. Cell24 should be set to switch off immediately after preamble and there is no change in cell power at step 1 as per test prose.

4. At step14a3 , attach reject cause should be ‘0F’ instead of ‘0D’ as per test prose.
5. Connection release should handled after sending attach reject.

6. TC should wait for trigger from EUTRA PTC to indicate end of test case

Summary of change
1. Initialised v_plmn after preamble.
2. Introduced a wait for trigger from EUTRA PTC before preamble in GERAN.
3. Cell power of cell24 is set to tsc_AttenuationCellOFF after preamble and removed the wait for trigger from EUTRA PTC.
4. At step14a3 , attach reject cause changed to ‘0F’. 

5. SS is now waiting for 250 ms for connection release indication from lower layer. 

6. introduced a wait for trigger from EUTRA PTC before end of  test case. 

Source of change
File : NAS_Attach_UG_GERAN.ttcn

Before:

Source: NAS_Attach_UG_GERAN.ttcn

function f_TC_9_2_1_2_13_GERAN () runs on GERAN_PTC

{

  var boolean v_TriggerRxd := false;

  var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

  var octetstring v_PLMN := f_GERAN_PLMN_Get(geran_Cell24);
  if (pc_GERAN and not pc_UTRA) { // UE supports GERAN but not UTRAN

    f_GERAN_Init(EUTRA_GERAN); // This is still only a 2 PTC test

    f_GERAN_CreateCell (geran_Cell24);

    f_GERAN_Pre_Preamble (geran_Cell24);

    f_GERAN_InterRAT_InitialiseAuthParams();  // The UE starts in EUTRA, so wait for it to send the Auth Params

    f_GERAN_TestBody_Set( true );

    // First Trigger

    //@siclog "Step 1" siclog@

    // Set Cell 24 to "Non Suitable Cell"

    v_TriggerRxd := f_WaitForIRATCoOrd_Trigger (EUTRA);
    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);

    // Second Trigger

    //@siclog "Step 14a1" siclog@

    // Set Cell 24 to "Serving Cell"

    v_TriggerRxd := f_WaitForIRATCoOrd_Trigger (EUTRA);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    //@siclog "Step 14a2" siclog@

    // Check: Does the UE transmit an ATTACH REQUEST message without P-TMSI, P-TMSI signature, RAI, TMSI

    f_UplinkTBFOnePhase (geran_Cell24);

    alt { // Receive GPRS Attach Request Message

      [] G_LLC.receive(car_G_LLC_UnitData_IND (cr_G_AttachReqWithIMSI(cr_AttachType(?, tsc_I_AttachTypeCombined),

                                                                      f_Imsi2MobileIdentity(px_IMSI_Def),

                                                                      cr_RAI (omit, cr_LAI(omit, v_PLMN, 'FFFE'O), ?),

                                                                      '111'B))) -> value v_LLCDataInd

        {

          f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Step 14a2");

        }

    }

    //@siclog "Step 14a3" siclog@

    // Send an ATTACH REJECT message with GMM Cause "Roaming not allowed in this location area"

     G_LLC.send(cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24), tsc_LLCSapi_GMM, tsc_LLC_PM, px_GSM_CipheringOnOff,

                           cs_G_AttachReject('0D'O))); // cause 'Roaming not allowed in this location area'

    //@siclog "Step 15" siclog@

    // Set Cell 24 to "Non Suitable Cell"

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);

    // Finished here for now

    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);

    // Test case now finished on GERAN side

    f_GERAN_TestBody_Set( false );

    f_GERAN_SS_CellRelease ( geran_Cell24 );

  } // UE doesn't support GERAN, or supports UTRAN too, so do nothing

} // f_TC_9_2_1_2_13_GERAN

After: NAS_Attach_UG_GERAN.ttcn
function f_TC_9_2_1_2_13_GERAN () runs on GERAN_PTC

{

  timer t_Dly := 0.250; //Anite change 4.5

  var boolean v_TriggerRxd := false;

  var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

  var octetstring v_PLMN ; //Anite change 4.1

  if (pc_GERAN and not pc_UTRA) { // UE supports GERAN but not UTRAN

    f_GERAN_Init(EUTRA_GERAN); // This is still only a 2 PTC test

    f_GERAN_CreateCell (geran_Cell24);

    v_PLMN := f_GERAN_PLMN_Get(geran_Cell24); //Anite change 4.1

    v_TriggerRxd := f_WaitForIRATCoOrd_Trigger (EUTRA); //Anite change 4.2

    f_GERAN_Pre_Preamble (geran_Cell24);

    f_GERAN_InterRAT_InitialiseAuthParams();  // The UE starts in EUTRA, so wait for it to send the Auth Params

    f_GERAN_TestBody_Set( true );

    // First Trigger

    //@siclog "Step 1" siclog@

    // Set Cell 24 to "Non Suitable Cell"

    //v_TriggerRxd := f_WaitForIRATCoOrd_Trigger (EUTRA); //anite change 4.3

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ChPwrLvl_Off); //Anite change 4.3

    // Second Trigger

    //@siclog "Step 14a1" siclog@

    // Set Cell 24 to "Serving Cell"

    v_TriggerRxd := f_WaitForIRATCoOrd_Trigger (EUTRA);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    //@siclog "Step 14a2" siclog@

    // Check: Does the UE transmit an ATTACH REQUEST message without P-TMSI, P-TMSI signature, RAI, TMSI

    f_UplinkTBFOnePhase (geran_Cell24);

    alt { // Receive GPRS Attach Request Message

      [] G_LLC.receive(car_G_LLC_UnitData_IND (cr_G_AttachReqWithIMSI(cr_AttachType(?, tsc_I_AttachTypeCombined),

                                                                      f_Imsi2MobileIdentity(px_IMSI_Def),

                                                                      cr_RAI (omit, cr_LAI(omit, v_PLMN, 'FFFE'O), ?),

                                                                      '111'B))) -> value v_LLCDataInd

        {

          f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Step 14a2");

        }

    }

    //@siclog "Step 14a3" siclog@

    // Send an ATTACH REJECT message with GMM Cause "Roaming not allowed in this location area"

     G_LLC.send(cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24), tsc_LLCSapi_GMM, tsc_LLC_PM, px_GSM_CipheringOnOff,

                           cs_G_AttachReject('0F'O))); //Anite change 4.4

    //UE will locally release the connection. SS should handle the release of TBF //Anite change 4.5

    t_Dly.start;

    alt {

      [] G_RLC.receive (car_G_RLC_ControlMsg_IND(geran_Cell24, tsc_PhyCh1, cr_PacketControlAcknowledgement))

        {

          t_Dly.stop;

        }

      [] t_Dly.timeout {}

    }
    //@siclog "Step 15" siclog@

    // Set Cell 24 to "Non Suitable Cell"

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);

    // Finished here for now

    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);

    v_TriggerRxd := f_WaitForIRATCoOrd_Trigger (EUTRA); //Anite change 4.6

    // Test case now finished on GERAN side

    f_GERAN_TestBody_Set( false );

    f_GERAN_SS_CellRelease ( geran_Cell24 );

  } // UE doesn't support GERAN, or supports UTRAN too, so do nothing

} // f_TC_9_2_1_2_13_GERAN

4.6 Change 5

Testcase name
9.2.1.2.13

Protocol
EMM

Function Name
TC_9_2_1_2_13

Reason for change
Guard timer value should be changed to a higher value to account for multiple plmn selection and multiple RAT selection

Summary of change
Guard timer changed to 960 seconds.  

Source of change
File : LTE_EPS_TS_Testcases_11wk04.ttcn

Before:

Source: LTE_EPS_TS_Testcases_11wk04.ttcn

  testcase TC_9_2_1_2_13() runs on MTC system SYSTEM {

    // @purpose

    //   Combined attach / Rejected / No suitable cells in tracking area

    var EUTRA_PTC   v_EUTRA     := EUTRA_PTC.create alive;

    var UTRAN_PTC   v_UTRAN     := UTRAN_PTC.create alive;

    var GERAN_PTC   v_GERAN     := GERAN_PTC.create alive;

    var CDMA2000_PTC v_CDMA2000  := null;

    timer t_GuardTimer := int2float(360);

    f_MTC_ConnectPTCs(v_EUTRA, v_UTRAN, v_GERAN, v_CDMA2000);

    v_EUTRA.start   (f_TC_9_2_1_2_13_EUTRA());

    v_UTRAN.start   (f_TC_9_2_1_2_13_UTRAN());

    v_GERAN.start   (f_TC_9_2_1_2_13_GERAN());

    t_GuardTimer.start;

    f_MTC_MainLoop(t_GuardTimer);

  }

After: LTE_EPS_TS_Testcases_11wk04.ttcn
  testcase TC_9_2_1_2_13() runs on MTC system SYSTEM {

    // @purpose

    //   Combined attach / Rejected / No suitable cells in tracking area

    var EUTRA_PTC   v_EUTRA     := EUTRA_PTC.create alive;

    var UTRAN_PTC   v_UTRAN     := UTRAN_PTC.create alive;

    var GERAN_PTC   v_GERAN     := GERAN_PTC.create alive;

    var CDMA2000_PTC v_CDMA2000  := null;

    timer t_GuardTimer := int2float(960);

    f_MTC_ConnectPTCs(v_EUTRA, v_UTRAN, v_GERAN, v_CDMA2000);

    v_EUTRA.start   (f_TC_9_2_1_2_13_EUTRA());

    v_UTRAN.start   (f_TC_9_2_1_2_13_UTRAN());

    v_GERAN.start   (f_TC_9_2_1_2_13_GERAN());

    t_GuardTimer.start;

    f_MTC_MainLoop(t_GuardTimer);

  }

5 Execution Log Files

5.1 Qualcomm 9600 UE 

The Qualcomm 9600 UE passed this test case on the Anite Conformance Toolset solution. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files tc_9_2_1_2_13_Qualcomm-log.html:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

6 References

[1]
R5s110159: This archive comprises html format execution log file.

